

Task №4

3D STRUCTURAL, THERMAL and THERMOELASTIC ANALYSES

KEYWORDS

1. 3D geometry: rotation, parallel transfer, extrusion
2. Static analysis
3. Structural analysis
4. Thermal analysis
5. Thermoelastic

PROBLEM DESCRIPTION

Let us consider three-dimensional domain obtained from a 2D-domain in the shape of the letter “R” by rotation around an axis. The geometry of the 2D-domain in Oxy coordinate system is shown in Fig. 1. The sizes of the letter are defined by the following parameters: $HL=1.4$ m (height); $WL=0.8$ m (width); $WWL=0.2$ m (width of the legs); $HAL=0.7$ m (additional height size). The volume (Fig. 2) is constructed by rotation of 2D-domain for an angle of 45° around a vertical axis $x=-WWL$. Material properties are as follows. The material of the domain is steel with the Young’s modulus $E=2\cdot 10^{11}$ N/m² and the Poisson’s ratio $\nu=0.29$, the thermal conductivity $k=46.7$ (W/(m·degree)) and the thermal extension coefficient $\alpha=1.51\cdot 10^{-5}$ 1/C.




Fig. 1. Scheme of the 2D-domain and axis of rotation Fig. 2. 3D-domain: rotation

The objective of the task is to perform structural, thermal and thermoelastic analyses.

Structural analysis. The bottom edges are rigidly fixed. The top edge is stretched by surface load $p=10$ GPa ($=10\cdot 10^9$ N/m²).

Thermal analysis. Boundary conditions: the bottom edges of the domain are subjected to heating with prescribed temperature values and the top edge is subjected to heat transfer. Let us assume that at the left bottom boundary the temperature θ_h is equal to 90°C ; and at the right bottom boundary the temperature θ_h is equal to 180°C . The heat transfer is set at

the top boundary with the value of the heat transfer coefficient 60 m^{-1} and the ambient temperature value 0°C . The rest of the boundary is thermally insulated.

Coupled thermoelastic analysis. The bottom edges are rigidly fixed and heated, the top edges are subjected only to heat transfer with no mechanical load. The objective is to calculate thermal stresses.

USING ANSYS TO SOLVE THE PROBLEM

All necessary parameters and materials properties are defined in the same way as it was for plane problems.

Here we concentrate on the commands that create a volume from a 2D-domain.

Rotation of an area around an axis

We need to create a volume by rotating an existing area (number 3) around the axis which is parallel to Oy -axis for an angle of 45° . This is done by **VROTAT** command. This command will create a new entity: volume (Fig. 2).

```
! Define keypoints for an axis of rotation (parallel to y-axis)
K,1001,-WWL,0
K,1002,-WWL,HL
```

```
! Create volume by rotating area 3 around the axis defined by
points 1001 and 1002 for an angle of 45 degrees
VROTAT,3,,,,,1001,1002,45
```

Drag of an area (parallel transfer) along a path

Here we consider a case when a given R-shaped area is dragged along z-axis (**VDRAG** command) for $4*WWL$ m. First we need to create a drag path which in this case will be a line located on the z-axis. This line is defined by two keypoints :

```
! Define keypoints and lines for a drag path
K,1001,0,0
K,1002,0,0,4*WWL
! New number for line numbering
NUMSTR,LINE,1001
L,1001,1002 ! Line 1001
! Create a volume by dragging area 3 along the created line
VDRAG,3,,,,,1001
```

Extrusion of an area (extruding and scaling of an area)

Here we consider a case when a given R-shaped area is extruded (**VEXT** command) along z-axis to $4*WWL$ m and scaled with the scale factor 0.5.

```
!Extrusion
!VEXT, NA1, NA2, NINC, DX, DY, DZ, RX, RY, RZ
! Create a volume by extruding area 3
VEXT,3,,,,, 4*WWL,0.5,0.5
```

3D finite element mesh

In order to build a finite element model, let us first define the element size.

```
! parameter for element size
dmesh=WWL/4
```

```
! set finite element size
esize,dmesh
```

With the help of KESIZE command we can set mesh condensation around keypoints (if necessary). LESIZE command sets the element size or the number of line divisions for a given line. For example:

```
! Divide line 1001 in DZ segments
LESIZE,1001,,,DZ
```

Then we choose the element type for our analysis and the desired type of the mesh. 3D finite elements can have various shapes, such as bricks, tetrahedrons, prisms. Figure below shows linear (without midside nodes) and quadratic (with midside nodes) elements.




Fig. 3. Finite elements types

Tetrahedrons with midside nodes can be used to mesh any volume of arbitrary shape (using VMESH command). Linear tetrahedrons (without midside nodes) are usually not recommended. Tables below summarize information on available finite elements for structural, thermal and thermoelastic analysis in Ansys 11.0.

Table 1. Elements for structural analysis (degrees of freedom: UX, UY in 2D; UX, UY, UZ in 3D)

Category	Element Name(s)
2-D Solids	PLANE25 , PLANE42 , PLANE82 , PLANE83 , VISCO88 , VISCO106 , VISCO108 , PLANE145 , PLANE146 , PLANE182 , PLANE183
3-D Solids	SOLID45 , SOLID46 , SOLID65 , VISCO89 , SOLID92 , SOLID95 , VISCO107 , SOLID147 , SOLID148 , SOLID185 , SOLID186 , SOLID187 , SOLID191 , SHELL281

Table 2.1 2-D thermal elements (degrees of freedom: TEMP)

Element	Dimens.	Shape or Characteristic	DOFs
PLANE35	2-D	Triangle, 6-node	Temperature (at each node)
PLANE55	2-D	Quadrilateral, 4-node	Temperature (at each node)
PLANE75	2-D	Harmonic, 4-node	Temperature (at each node)
PLANE77	2-D	Quadrilateral, 8-node	Temperature (at each node)

Element	Dimens.	Shape or Characteristic	DOFs
PLANE78	2-D	Harmonic, 8-node	Temperature (at each node)

Table 2.2 3-D thermal elements (degrees of freedom: TEMP)

Element	Dimens.	Shape or Characteristic	DOFs
SOLID70	3-D	Brick, 8-node	Temperature (at each node)
SOLID87	3-D	Tetrahedron, 10-node	Temperature (at each node)
SOLID90	3-D	Brick, 20-node	Temperature (at each node)

Table 3. Finite elements for structural-thermal (thermoelastic) analysis

Elements	Modeled effects	Analysis types
Linear elements SOLID5 - Coupled-Field Hexahedral PLANE13 - Coupled-Field Quadrilateral SOLID98 - Coupled-Field Tetrahedral	Thermoelastic (Thermal Stress)	Static Full Transient
Quadratic elements PLANE223 - Coupled-Field Quadrilateral SOLID226 - Coupled-Field Hexahedral SOLID227 - Coupled-Field Tetrahedral	Thermoelastic (Thermal Stress and Piezocaloric) Structural material nonlinearities Thermoplastic	Static Full Harmonic Full Transient

1) Hexahedral meshing

In this task, we will learn how to create a hexahedral mesh for a volume that follows a certain pattern. Here, we will use 8-node (linear) hexahedrons. Such elements are SOLID70 for thermal analysis and SOLID45 for structural analysis. With ET command we choose the solid element type (number 1).

```
ET,1,SOLID70 ! 3-D 8 node thermal solid, degree of freedom: TEMP
              (temperature)
! ET,1,SOLID45 ! 3-D 8 node structural solid, degrees of freedom:
ux,uy,uz (displacements)
```

In ANSYS, 11 node tetrahedrons (with midside nodes) are SOLID87 for thermal analysis and SOLID187 for structural analysis.

Let us note that the 3D letter R can be meshed with hexahedrons by sweeping a 2D quadrilateral mesh through a volume (obtaining hexahedrons from quadrilaterals) because this is the way the volume was created. We introduce an auxiliary geometry-only finite element **MESH200**, that does not have degrees of freedom. This element will set the pattern for 3D-mesh. The shape of this element is 4-node quadrilateral: keyopt(1)=6 With ET command we define this element type (number 2).

```
ET,2,MESH200,6 ! FE of type 2: auxiliary mesh-only element, no de-
degrees of freedom
```




Fig. 5. Geometry of finite element MESH200

After that we mesh the source area with quadrilaterals. First, we set element type 2, then select the plane area and mesh it with the help of AMESH command. Default mesh type is free mesh:

```
TYPE,2 ! set element type 2 (to mesh 2D domain)
! Select existing 2D-domain
ASEL,S,LOC,Z,0
! Mesh 2D-domain with auxiliary elements to define mesh pattern
AMESH,ALL
```

3D-mesh is created by sweeping the 2D-mesh through the volume using **VSWEEP** command (it is necessary to first choose the element type number 1).

```
TYPE,1 ! set element type 1 (to mesh a volume), main FEF
! Fill an existing unmeshed volume with elements by sweeping the
mesh from an adjacent area through the volume
VSWEEP,ALL
```

After that we select all areas again:

```
Asel,all ! select all areas
```

The resulting finite element model for a volume created by rotation is shown in Fig. 3. Fig.4 illustrates a volume created by dragging an area along z-axis (parallel transfer). In order to plot elements, use **EPLLOT** command or Menu path: Plot->Elements.




Fig. 1. Hexahedral finite element mesh: rotation




Fig. 2. Hexahedral finite element mesh: parallel transfer (drag along z-axis)

2) Tetrahedral meshing

Any volume of arbitrary shape can be always with tetrahedrons. Free tetrahedral mesh is created by standard **VMESH** command, for example:

```
ET,1,SOLID87 ! 3-D 8 node thermal solid, degree of freedom: TEMP
              (temperature)
VMESH,ALL ! Mesh all volumes
```

The resulting finite element tetrahedral mesh for a volume created by rotation is shown in Fig. 5.




Fig. 5. Tetrahedral finite element mesh: rotation

Boundary conditions and solution

The next step is to set boundary conditions and solve the system of finite element equations. The actions can be done in the solver. **/SOLU** command enters the solver.

As example, let us consider the boundary conditions for thermal analysis in a thermal conductivity problem

Case of thermal conductivity problem: only thermal analysis

```
! Boundary conditions for thermal analysis
TREF,0 ! Define reference temperature
! Define new cylindrical coordinate system with z-axis coincident
with the axis of rotation (to further select the nodes)

LOCAL,12,1,-WWL,0,0,, -90
CSYS,12
NSEL,S,LOC,Z,0 ! Select all nodes with Z''=0
NSEL,R,LOC,X,WWL,2*WWL
D,ALL,TEMP,T_INP1 ! For all selected nodes define TEMP=T_INP1
NSEL,S,LOC,Z,0 ! Select all nodes with Z''=0
NSEL,R,LOC,X,WL+WWL,WL
D,ALL,TEMP,T_INP2 ! For all selected nodes define TEMP=T_INP2
NSEL,S,LOC,Z,HL,HL !Select all nodes with Z''=HL
NSEL,R,LOC,X,WWL,2*WWL
SF,ALL,CONV,H_F,T_EXT ! For all selected nodes define convective
heat transfer
NSEL,ALL ! Select all nodes
CSYS,0
```

Let us give some comments to the commands that are used to set boundary conditions. In order to get access to the necessary nodes (for thermal boundary conditions) here it can be convenient to move on to the cylindrical coordinate system Ox''y''z'', which axis z'' will be the axis of rotation for original plane domain. Therefore we used **LOCAL** command to introduce new cylindrical coordinate system under number 12 displaced by WWL to the left of the original y-axis along the x-axis and rotated by 90° with respect to x-axis in the direction of rotation from z-axis to y-axis (rotation angle THYZ = -90°). After changing active coordinate system to the user system 12 (**CSYS,12**) **NSEL** command works so that x-axis represents r-axis of the current coordinate system.

Another way to select necessary nodes or elements for boundary conditions is to first select areas containing these nodes, using commands **NSLA** or **ESLA** (for elements). Here is an example for convective heat transfer on the area 9.

```
ASEL,S,AREA,,9
NSLA,S,1 ! Select nodes belonging to the selected areas, including
the area boundary
! Convective heat transfer in the selected nodes
SF,ALL,CONV,H_F,T_EXT
! select everything
ALLSEL
```

Temperature in selected nodes is set by **D** command. Convective heat transfer in selected nodes is set by **SF** command. The use of these commands is the same in 2D and 3D.

Case of linear elasticity problem: only structural analysis

For structural analysis **D** command sets displacements in selected nodes, whereas **SF** commands defines distributed pressure load.

It is also possible to set boundary conditions on selected areas, using corresponding **DA** and **SFA** commands.

Case of thermoelastic problem: coupled thermal-structural analysis

Thermoelastic analysis requires to set additional material properties, such as the density and the thermal expansion coefficient.

```
!-----
! "Thermal" material properties
T_INP1=90    ! Temperature at the left bottom edge
T_INP2=180   ! Temperature at the right bottom edge
T_EXT=0     ! Temperature of external medium T_EXT=0
H_F=60      ! Convective heat transfer coefficient

MP,KXX,1,46.7 ! Thermal conductivity coefficient KXX=46.7 (Steel)

ALPH1=1.51e-05 ! Thermal expansion ratio
MP,ALPX,1,ALPH1
!-----
! "Elastic" material properties
! Material properties of steel
RO1=7.8e3 ! density
E1=2.1e11 ! Young's modulus
NU1=0.29 ! Poisson's ratio

MP,DENS,1,RO1
MP,EX,1,E1
MP,NUXY,1,NU1
!-----
```

All mechanical and thermal boundary conditions are set in usual way:

```
! Решение связанной задачи термоупругости
/SOLU
ANTYPE,STAT ! Static analysis

! Boundary conditions for thermal analysis
TREF,0 ! Define reference temperature
! Define new cylindrical coordinate system with z-axis coincident
with the axis of rotation (to further select the nodes)

LOCAL,12,1,-WWL,0,0,, -90
CSYS,12
NSEL,S,LOC,Z,0 ! Select all nodes with Z''=0
NSEL,R,LOC,X,WWL,2*WWL
D,ALL,TEMP,T_INP1 ! For all selected nodes define TEMP=T_INP1
NSEL,S,LOC,Z,0 ! Select all nodes with Z''=0
NSEL,R,LOC,X,WL+WWL,WL
D,ALL,TEMP,T_INP2 ! For all selected nodes define TEMP=T_INP2
NSEL,S,LOC,Z,HL,HL !Select all nodes with Z''=HL
NSEL,R,LOC,X,WWL,2*WWL
SF,ALL,CONV,H_F,T_EXT ! For all selected nodes define convective
heat transfer
NSEL,ALL ! Select all nodes
CSYS,0

! Boundary conditions for structural analysis
```



```

ASEL,S,AREA,,1,5,4
NSLA,S,1
D,ALL,UX,0
D,ALL,UY,0
D,ALL,UZ,0
ASEL,ALL
NSEL,ALL      ! Select all nodes

SOLVE          ! Solve the problem
FINISH

```

RESULTS FOR THERMOELASTIC ANALYSIS

The finite element model with applied boundary conditions for thermoelastic analysis for a 3D R-shaped domain obtained by rotation is shown in Fig. 6. (Menu path: Plot->Elements, for showing boundary conditions go to PltCtrls->Symbols->tick All applied BC, select showing convective heat transfer condition: Surface Load Symbols->Convective FilmCoef):





Fig. 6 Finite element model with boundary conditions (thermoelastic analysis)

A deformed shape of the domain is shown in Fig. 7. The deformation occurs due to thermal stresses.


DISPLACEMENT
 STEP=1
 SUB =1
 TIME=1
 DMX =.004448



Thermal stress in 3D region (R)

Fig. 7 Deformed finite element mesh and undeformed edge (thermoelastic analysis)

Results on temperature and thermal flux distribution are shown in Fig. 8 and 9.



Thermal stress in 3D region (R)

Fig. 8 Temperature distribution in 3D domain (thermoelastic analysis)




Fig. 9 Thermal flux distribution in 3D domain (thermoelastic analysis)

Fig. 10 and 11 illustrate the distribution of the displacement vector magnitude and stress intensity.




Fig. 10 Distribution of the displacement vector magnitude




Fig. 11 Distribution of the stress intensity